Lab 5 Linear Regression with Within-subject Correlation

<u>Goals</u>:

- Fit linear regression models that account for within-subject correlation using Stata.
- Compare weighted least square, GEE, and random effect modeling.
- Compare correlation specification
- Interpret model coefficients

<u>Data</u>:

Use the pig data which is in wide format:

```
. ** clear any existing data **
. clear
. ** increase the memory **
. set memory 40m
. ** increase the number of variables you can include in the dataset **
. set matsize 500

. ** read in the data set **
. ** Make sure you specify the path correctly! **
. use pigs.stata.dta, clear

. ** alternatively click on: file>open>pigs.stata.dta **
. ** reshape to long **

. reshape long week, i(Id) j(time)
. rename week weight

. ** sort data appropriately **
. sort Id time

. ** spaghetti plot data **
. twoway line weight time, c(L)
```



Part A. Ordinary Least Squares (OLS) by hand using Stata matrix functions

We wish to fit a model given by:

$$Weight_{ij} = \beta_0 + \beta_1\ Time_{ij} + \varepsilon_{ij} \qquad \varepsilon_{ij} \sim N(0, \sigma^2)$$

where i = pig ID and j = time.

```
. gen intercept=1
. mkmat intercept time, matrix(X)

. **look at which matrices Stata has defined **
. matrix dir
           X[432,2]

. ** read in outcome variable as a vector **
. mkmat weight, matrix(Y)

. ** use definition of OLS estimates to get parameter estimates **
. matrix betahatOLS = invsym(X'*X)*X'*Y

. ** look at your estimates **
. matrix list betahatOLS

  betahatOLS[2,1]
  weight
  intercept  19.35561
  time  6.2098958
```

Check and see if your estimates are the same as Stata:

```
. regress weight time

      Source |       SS       df       MS              Number of obs =     432
-------------+------------------------------           F(  1,   430) = 5757.41
       Model |  111060.882     1  111060.882           Prob > F      =  0.0000
    Residual |  8294.72677   430  19.2900622           R-squared     =  0.9305
-------------+------------------------------           Adj R-squared =  0.9303
       Total |  119355.609   431  276.927167           Root MSE      =   4.392

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.209896   .0818409    75.88   0.000     6.049038    6.370754
       _cons |   19.35561   .4605447    42.03   0.000     18.45041    20.26081
------------------------------------------------------------------------------
```

Part B. Weighted Least Squares (WLS)

!!!WARNING!!!
We can find wls in STATA in the options of "xtreg, be".
What does "xtreg" do? Search in the STATA help file.

```
. ** Try the be and wls option **
. xtreg weight time, be wls i(Id)

Between regression (regression on group means)  Number of obs      =       432
Group variable: Id                              Number of groups   =        48

R-sq:  within  =      .                         Obs per group: min =         9
       between = 0.0000                                        avg =       9.0
       overall = 0.0000                                        max =         9

                                                F(0,47)            =      0.00
sd(u_i + avg(e_i.))=  3.953498                  Prob > F           =         .

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |  (dropped)
       _cons |   50.40509   .5706383    88.33   0.000     49.25712    51.55307
------------------------------------------------------------------------------
```

What model did Stata fit?

Stata doesn't really have a generic WLS command. To fit WLS ourselves, we first get a uniform (exchangeable) correlation estimate, using the GEE routine for convenience. Another possibility is to average all the correlations in the ACF.

```
. ** the quietly option hides the results from the output window **
. quietly xtgee weight time, i(Id) corr(exch)

. xtcorr

Estimated within-Id correlation matrix R:
c1 c2 c3 c4 c5 c6 c7 c8 c9
r1 1.0000
r2 0.7717 1.0000
r3 0.7717 0.7717 1.0000
r4 0.7717 0.7717 0.7717 1.0000
r5 0.7717 0.7717 0.7717 0.7717 1.0000
r6 0.7717 0.7717 0.7717 0.7717 0.7717 1.0000
r7 0.7717 0.7717 0.7717 0.7717 0.7717 0.7717 1.0000
r8 0.7717 0.7717 0.7717 0.7717 0.7717 0.7717 0.7717 1.0000
r9 0.7717 0.7717 0.7717 0.7717 0.7717 0.7717 0.7717 0.7717 1.0000
```

So a good estimate of rho is 0.7717

<u>Create WLS weighting matrix by hand.</u>

Use uniform correlation formula: Corr = (1-rho)I + rho(11')

```
. ** Using Stata matrix commands **
```

```
. ** This is one way to define a number in Stata, since the **
. ** STATA dataset only deals with variables **

. matrix rho=0.7717

. matrix list rho

symmetric rho[1,1]
c1
r1 .7717

. ** create a vector of 1s, with the length as the number of **
. ** observations within each subject, here is 9 **
. matrix one = (1\1\1\1\1\1\1\1\1)
. matrix list one

one[9,1]
c1
r1 1
r2 1
r3 1
r4 1
r5 1
r6 1
r7 1
r8 1
r9 1

. ** create the matrix of all 1s **
. matrix oneonet = one*(one')
. matrix list oneonet

symmetric oneonet[9,9]
      r1 r2 r3 r4 r5 r6 r7 r8 r9
      r1 1
      r2 1 1
      r3 1 1 1
      r4 1 1 1 1
      r5 1 1 1 1 1
      r6 1 1 1 1 1 1
      r7 1 1 1 1 1 1 1
      r8 1 1 1 1 1 1 1 1
      r9 1 1 1 1 1 1 1 1 1

** "symmetric" means it is a symmetric matrix, and only half of the
** values are displayed. **
** create the identity matrix using the I() command, **
. matrix I = I(9)
. matrix list I

symmetric I[9,9]
      c1 c2 c3 c4 c5 c6 c7 c8 c9
      r1 1
      r2 0 1
      r3 0 0 1
      r4 0 0 0 1
      r5 0 0 0 0 1
      r6 0 0 0 0 0 1
      r7 0 0 0 0 0 0 1
```

```
    r8 0 0 0 0 0 0 0 0 1
    r9 0 0 0 0 0 0 0 0 0 1

. ** create the correlation matrix according to the fomular **
. matrix R = (1-rho)*I + rho*oneonet
. matrix list R

symmetric R[9,9]
r1 r2 r3 r4 r5 r6 r7 r8 r9
r1 1
r2 .7717 1
r3 .7717 .7717 1
r4 .7717 .7717 .7717 1
r5 .7717 .7717 .7717 .7717 1
r6 .7717 .7717 .7717 .7717 .7717 1
r7 .7717 .7717 .7717 .7717 .7717 .7717 1
r8 .7717 .7717 .7717 .7717 .7717 .7717 .7717 1
r9 .7717 .7717 .7717 .7717 .7717 .7717 .7717 .7717 1
```

Trick: use xtgee command to do the WLS with the created correlation matrix.
But GEE and WLS are not equivalent.

```
. xtgee weight time, i(Id) t(time) corr(fixed R)


Iteration 1: tolerance = 7.330e-15

GEE population-averaged model                   Number of obs      =       432
Group and time vars:                Id time     Number of groups   =        48
Link:                              identity     Obs per group: min =         9
Family:                            Gaussian                    avg =       9.0
Correlation:                fixed (specified)                  max =         9
                                                Wald chi2(1)       =  25335.93
Scale parameter:                   19.20076     Prob > chi2        =    0.0000

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.209896   .0390136   159.17   0.000     6.133431    6.286361
       _cons |   19.35561   .5974033    32.40   0.000     18.18472     20.5265
------------------------------------------------------------------------------
```

WLS and GEE are for marginal models, the estimated coefficients correspond to population-average effect.

Although the estimated coefficients are the same as the model without considering with-subject correlation, the standard errors change, our inference changes.

Try WLS with another uniform correlation, like 0.5 for fun!

```
. matrix rho=0.5
. matrix one = (1\1\1\1\1\1\1\1\1)
. matrix oneonet = one*(one')
. matrix I = I(9)
. matrix R = (1-rho)*I + rho*oneonet
```

```
. matrix list R
symmetric R[9,9]
r1 r2 r3 r4 r5 r6 r7 r8 r9
r1 1
r2 .5 1
r3 .5 .5 1
r4 .5 .5 .5 1
r5 .5 .5 .5 .5 1
r6 .5 .5 .5 .5 .5 1
r7 .5 .5 .5 .5 .5 .5 1
r8 .5 .5 .5 .5 .5 .5 .5 1
r9 .5 .5 .5 .5 .5 .5 .5 .5 1
```

Do WLS with new weighting (correlation) matrix and compare:

```
. xtgee weight time, i(Id) t(time) corr(fixed R)

Iteration 1: tolerance = 1.047e-14

GEE population-averaged model                   Number of obs      =       432
Group and time vars:                Id time      Number of groups   =        48
Link:                              identity      Obs per group: min =         9
Family:                            Gaussian                     avg =       9.0
Correlation:              fixed (specified)                     max =         9
                                                 Wald chi2(1)       =  11568.39
Scale parameter:                   19.20076      Prob > chi2        =    0.0000

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.209896   .0577362   107.56   0.000     6.096735    6.323057
       _cons |   19.35561   .5527817    35.01   0.000     18.27218    20.43905
------------------------------------------------------------------------------
```

The estimated coefficients are still the same, the standard errors change.

Part C. Comparison of the correlation structures: independent, exchangeable, ARI, and unstructured.

Correlation Structure	Model
Independent	OLS
Exchangeable	GEE and Random Effect Model
ARI	GEE and Random Effect Model
Unstructured	GEE

I. Independent correlation structure:

(1) Ordinary Least Squares

```
. regress weight time

      Source |       SS       df       MS              Number of obs =     432
-------------+------------------------------           F(  1,   430) = 5757.41
       Model |  111060.882     1  111060.882           Prob > F      =  0.0000
    Residual |  8294.72677   430  19.2900622           R-squared     =  0.9305
-------------+------------------------------           Adj R-squared =  0.9303
       Total |  119355.609   431  276.927167           Root MSE      =   4.392

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.209896   .0818409    75.88   0.000     6.049038    6.370754
       _cons |   19.35561   .4605447    42.03   0.000     18.45041    20.26081
------------------------------------------------------------------------------
. predict wtpred
. twoway line weight wtpred time ,c(L) clwidth(thin thick)
```




```
. drop wtpred
```

(2) GEE with independent correlation:

```
. xtgee weight time, i(Id) corr(ind)

Iteration 1: tolerance = 1.971e-15

GEE population-averaged model                   Number of obs      =       432
Group variable:                         Id      Number of groups   =        48
Link:                             identity      Obs per group: min =         9
Family:                           Gaussian                     avg =       9.0
Correlation:                   independent                     max =         9
                                                Wald chi2(1)       =   5784.19
Scale parameter:                  19.20076      Prob > chi2        =    0.0000

Pearson chi2(432):                 8294.73      Deviance           =   8294.73
Dispersion (Pearson):             19.20076      Dispersion         =  19.20076

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.209896   .0816513    76.05   0.000     6.049862    6.369929
       _cons |   19.35561   .4594773    42.13   0.000     18.45505    20.25617
------------------------------------------------------------------------------
. xtcorr

Estimated within-Id correlation matrix R:

        c1      c2      c3      c4      c5      c6      c7      c8      c9
r1  1.0000
r2  0.0000  1.0000
r3  0.0000  0.0000  1.0000
r4  0.0000  0.0000  0.0000  1.0000
r5  0.0000  0.0000  0.0000  0.0000  1.0000
r6  0.0000  0.0000  0.0000  0.0000  0.0000  1.0000
r7  0.0000  0.0000  0.0000  0.0000  0.0000  0.0000  1.0000
r8  0.0000  0.0000  0.0000  0.0000  0.0000  0.0000  0.0000  1.0000
r9  0.0000  0.0000  0.0000  0.0000  0.0000  0.0000  0.0000  0.0000  1.0000
```

We can see that they give the same results. The slight difference in the standard error of the coefficients should come from that GEE needs iteration for estimation.

Alternative commands (`xtreg, pa)` is just GEE estimate.

II. Exchangeable (uniform) correlation structure:

Here we assume the correlation between measurements within the same pig Cor(Y_{ij}, $Y_{ij'}$) is a constant. Specifically, this correlation is independent of the time between measurements.

(1) Marginal specification of uniform correlation via GEE:

GEE is one method to analyze longitudinal data when marginal correlation structure is specified.

We wish to fit the model:

$$Weight_{ij} = \beta_0 + \beta_1 \; Time_{ij} + \varepsilon_{ij} \qquad \varepsilon_{ij} \sim N(\,0, \sigma^2) \quad cor(\varepsilon_{ij}\,, \varepsilon_{ij'}) = \rho$$

Alternatively:

$$Var\left(Weight_i\right) = \sigma^2 \begin{pmatrix} 1 & \rho & \rho & \rho \\ \rho & 1 & \vdots & \rho \\ \rho & \dots & 1 & \rho \\ \rho & \rho & \rho & 1 \end{pmatrix}$$

```
. xtgee weight time, i(Id) corr(exch)

Iteration 1: tolerance = 5.585e-15
GEE population-averaged model                   Number of obs      =       432
Group variable:                         Id      Number of groups   =        48
Link:                             identity      Obs per group: min =         9
Family:                           Gaussian                     avg =       9.0
Correlation:                  exchangeable                     max =         9
                                                Wald chi2(1)       =  25337.48
Scale parameter:                  19.20076      Prob > chi2        =    0.0000

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.209896   .0390124   159.18   0.000     6.133433    6.286359
       _cons |   19.35561   .5974055    32.40   0.000     18.18472    20.52651
```

```
. ** Alternative to fit marginal uniform corr with GEE **
. xtreg weight time, pa i(Id) corr(exch)
```

To look at the estimated correlation matrix:

```
. xtcorr
Estimated within-Id correlation matrix R:
c1 c2 c3 c4 c5 c6 c7 c8 c9
r1 1.0000
r2 0.7717 1.0000
r3 0.7717 0.7717 1.0000
r4 0.7717 0.7717 0.7717 1.0000
```

Plot of the predicted growth trend:

```
. xtreg weight time, pa i(Id) corr(exch)
. predict wtpred
. twoway line weight wtpred time ,c(L) clwidth(thin thick)
```




```
. drop wtpred
```

We can see it is the same as the one from independent correlation structure. This is because in linear regression, the estimate of the mean function coefficient is independent of the estimate of the variance and the correlation structure.

(2) Conditional specification of uniform correlation using random effect model:

Random effect model specifies the correlation structure through the conditional distribution, conditioned on the random effects *U_i*. We can derive the marginal correlation structure from the conditional distribution. The marginal correlation structure induced from random intercept model is exchangeable.

We wish to fit the model:

$$Weight_{ij} = \beta_0 + u_i + \beta_1\, Time_{ij} + \varepsilon_{ij} \qquad u_i \sim N(0, \sigma_u^2) \qquad \varepsilon_{ij} \sim N(0, \sigma_\varepsilon^2)$$

This gives the same variance structure

$$Var(Weight_i) = (\sigma_\tau^{\,2} + \sigma_u^{\,2})\begin{pmatrix} 1 & \rho & \rho & \rho \\ \rho & 1 & \vdots & \rho \\ \rho & \dots & 1 & \rho \\ \rho & \rho & \rho & 1 \end{pmatrix} \quad \text{where} \quad \rho = \frac{\sigma_u^2}{\sigma_\tau^2 + \sigma_u^2}$$

The Stata command `xtreg` allow two methods to fit the random effect model: GLS and MLE.

```
. xtreg weight time, re i(Id)

Random-effects GLS regression                   Number of obs      =       432
Group variable: Id                              Number of groups   =        48

R-sq:  within  = 0.0000                         Obs per group: min =         9
       between = 0.0000                                        avg =       9.0
       overall = 0.9305                                        max =         9

Random effects u_i ~ Gaussian                   Wald chi2(1)       =  25271.50
corr(u_i, X)       = 0 (assumed)                Prob > chi2        =    0.0000

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.209896   .0390633   158.97   0.000     6.133333    6.286458
       _cons |   19.35561    .603139    32.09   0.000     18.17348    20.53774
-------------+----------------------------------------------------------------
     sigma_u |  3.8912528
     sigma_e |  2.0963561
         rho |  .77505203   (fraction of variance due to u_i)

. xtreg weight time, re i(Id) mle

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.209896   .0390124   159.18   0.000     6.133433    6.286359
       _cons |   19.35561   .5974055    32.40   0.000     18.18472    20.52651
-------------+----------------------------------------------------------------
    /sigma_u |    3.84935   .4058114                      3.130767    4.732863
    /sigma_e |   2.093625   .0755471                       1.95067    2.247056
         rho |    .771714   .0393959                      .6876303    .8413114
------------------------------------------------------------------------------
Likelihood-ratio test of sigma_u=0: chibar2(01)=  472.65 Prob>=chibar2 = 0.000
```

The MLE method for fitting random effect model can also be obtained by the following command:

```
. xtreg weight time, i(Id) mle
```

It performs exactly the same procedure as the previous command.

Plot the predicted growth trend from the random effect model:

```
. quietly xtreg weight time, re i(Id) mle
. ** predict the marginal mean function **
. predict wtpred
(option xb assumed; fitted values)

.** predict the individual pig mean function **
. predict wtpredi, xbu
```

```
. twoway line weight wtpredi wtpred time,c(L)
```




```
. drop wtpred wtpredi
```

It can be seen that the marginal mean growth trend which is averaged from the individual growth trend is the same as the one estimated using marginal specification of the correlation structure.

III. AR1 correlation structure:

(1) Marginal specification of AR1 correlation structure:

```
. xtgee weight time, i(Id) corr(AR1) t(time)

Iteration 1: tolerance = .02513276
Iteration 2: tolerance = .00009237
Iteration 3: tolerance = 4.366e-07

GEE population-averaged model                   Number of obs      =       432
Group and time vars:                   Id time   Number of groups   =        48
Link:                                 identity   Obs per group: min =         9
Family:                               Gaussian                  avg =       9.0
Correlation:                             AR(1)                  max =         9
                                                Wald chi2(1)       =   6254.91
Scale parameter:                      19.26754   Prob > chi2        =    0.0000

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.272089   .0793052    79.09   0.000     6.116654    6.427524
       _cons |   18.84218   .6745715    27.93   0.000     17.52004    20.16431
------------------------------------------------------------------------------
```

```
. xtcorr

Estimated within-Id correlation matrix R:
c1 c2 c3 c4 c5 c6 c7 c8 c9
r1 1.0000
r2 0.9167 1.0000
r3 0.8403 0.9167 1.0000
r4 0.7702 0.8403 0.9167 1.0000
r5 0.7061 0.7702 0.8403 0.9167 1.0000
r6 0.6472 0.7061 0.7702 0.8403 0.9167 1.0000
r7 0.5933 0.6472 0.7061 0.7702 0.8403 0.9167 1.0000
r8 0.5438 0.5933 0.6472 0.7061 0.7702 0.8403 0.9167 1.0000
r9 0.4985 0.5438 0.5933 0.6472 0.7061 0.7702 0.8403 0.9167 1.0000
```

The AR parameter is 0.9167, the correlation with time lag 1.

(2) Conditional specification of AR1 correlation using random effect model:

```
** xtregar cannot use the variable specificed for time as covariate in
** regression.
** generate variable "week" for the tsset command **
. gen week=time
. xtset Id week
panel variable: Id, 1 to 48
time variable: week, 1 to 9

. xtregar weight time

RE GLS regression with AR(1) disturbances       Number of obs      =       432
Group variable: Id                              Number of groups   =        48

R-sq:  within  = 0.9851                         Obs per group: min =         9
       between = 0.0000                                        avg =       9.0
       overall = 0.9305                                        max =         9

                                                Wald chi2(2)       =  12688.55
corr(u_i, Xb)      = 0 (assumed)                Prob > chi2        =    0.0000

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.257651   .0555527   112.64   0.000      6.14877    6.366533
       _cons |   19.00945   .6281622    30.26   0.000     17.77827    20.24062
-------------+----------------------------------------------------------------
      rho_ar |  .73091237   (estimated autocorrelation coefficient)
     sigma_u |   3.583343
     sigma_e |  1.5590851
     rho_fov |  .84082696   (fraction of variance due to u_i)
       theta |  .60838037
------------------------------------------------------------------------------
```

We can see the results are different from the marginal specification, in both the standard error of beta and in the AR parameter.

```
. ** predict the marginal mean trend **
. predict wtpred
. ** predict the random intercept **
. predict ui, u
(48 missing values generated)
(48 missing values generated)
. ** generate the individual mean trend **
```

```
. gen wtpredi = wtpred + ui
(48 missing values generated)
. twoway (line weight time, c(L)) (line wtpredi time, c(L) ) (line
wpred time, lwidth(thick))
```




```
. drop wtpred wtpredi u
```

The individual growth trend is different from the ones with uniform correlation. This is because we give different assumption for the individual random intercept.

IV. Unstructured correlation:

All the above are parametric methods, because we make assumptions for the form of the correlation structure. The unstructured methods is non-parametric, therefore it will reflect the actual correlation structure in the data. However, it has the drawback that it takes away too much of the degree of freedom in the model fitting, since every single element in the correlation matrix is estimated as a parameter.

The guideline would be: when the number of observations for each subject is not big, i.e. the correlation matrix is not in large dimension, it is suggested to compare the unstructured inference with other parametric inference. If they are close, then we used the parametric inference because they are more statistically powerful and does not lose much information. If they are not close, unstructured inference should be used. However, when the correlation matrix is too large, we have to use parametric methods.

```
. xtgee weight time, i(Id) corr(unstr) t(time)


Iteration 1: tolerance = .01589005
Iteration 2: tolerance = .00153151
Iteration 3: tolerance = .00027435
Iteration 4: tolerance = .0000591
Iteration 5: tolerance = .00001379
Iteration 6: tolerance = 3.304e-06
Iteration 7: tolerance = 7.981e-07

GEE population-averaged model                   Number of obs      =       432
Group and time vars:                 Id time    Number of groups   =        48
Link:                               identity    Obs per group: min =         9
Family:                             Gaussian                   avg =       9.0
Correlation:                    unstructured                   max =         9
                                                Wald chi2(1)       =   3271.77
Scale parameter:                     19.2616    Prob > chi2        =    0.0000

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        time |   6.120706   .1070065    57.20   0.000     5.910977    6.330434
       _cons |   19.71316   .5242296    37.60   0.000     18.68569    20.74063

working correlation matrix not positive definite
convergence not achieved

. xtcorr


        c1      c2      c3      c4      c5      c6      c7      c8      c9
r1  1.0000
r2  0.3280  1.0000
r3  0.3232  0.4513  1.0000
r4  0.3646  0.4794  0.6529  1.0000
r5  0.4333  0.5681  0.7516  0.8266  1.0000
r6  0.3933  0.5271  0.7266  0.7880  0.9457  1.0000
r7  0.4131  0.5483  0.7517  0.8186  0.9806  1.0000  1.0000
r8  0.3995  0.5432  0.8237  0.8604  1.0000  1.0000  1.0000  1.0000
r9  0.4261  0.5928  0.8948  0.9492  1.0000  1.0000  1.0000  1.0000  1.0000
```

We can see the estimated correlation matrix is neither uniform or AR1. More importantly, the standard error for betas are very different from the ones with uniform and AR1 correlation structure. So should use unstructured.

Q1 Does this look stationary?
No!! Correlations with the same time lag are different. So the correlation does not depend on the length of lag only.

Q2 How can we program an "unstructured" correlation structure with random effects?
No, you couldn't, random effects model is completely parametric.

Finally, a comparison of regression coefficients and standard errors:

Cor Structure		**Model**	**Time**	**se (Time)**
Independence	OLS		6.21	0.081
Exchangeable	WLS	rho = 0.77	6.21	0.039
		rho = 0.5	6.21	0.058
	GEE	rho = 0.77	6.21	0.039
	RE	rho = 0.77	6.21	0.039
AR1	GEE	AR = 0.91	6.27	0.079
	RE	AR = 0.73	6.25	0.055
Unstructured	GEE		6.12	0.107